# Mathematically designing an egg

K. Foutzopoulos[a], 27 April 2021

```
(%i1) kill(all)$
(%i1) ratprint: false$
```

Eggs are oval shaped objects with their form varying among species in symmetry and ellipticity. An oval[1] is loosely defined as a squashed ellipse that usually (not required) is pointier at one end, that is it has only one symmetry axis. This article will attempt to intuitively design a few such shapes, utilizing the computer algebra system Maxima.

Before proceeding some good resources worth taking a look. Crespi & You[2] have made an interactive page on Science magazine showcasing the egg variety. The site Mathematische Basteleien has a page[3] containing numerous ovals and egg curves. The basis of this article is formed by utilizing the pertubing functions found there. Another extensive source is a series of articles[4], similar in feel to this, published by Nobuo Yamamoto.

A simple way to develop an oval is by perturbing an ellipse in a way that one of the two coordinates has a different size at the two half-planes. An ellipse[5] can be defined implicitly as

$$x^2/a^2 + y^2/b^2 = 1$$

An horizontal oval (only symmetry axis is $x$-axis) can then be described

$$x^2/a^2 + y^2 t(x)/b^2 = 1 \quad a > b$$

Some example forms the perturbing function $t(x)$ can take are $1 + qx$, $(1 - qx)^{-1}$, and $e^{qx}$. Plotting (setting for example $q = 0.2$) them it can be seen (alternatively computing that $t'(x) > 0$) that

$$\forall \xi > 0 \colon |t(\xi)| \geqslant |t(-\xi)| \Rightarrow |y(\xi)| \leqslant |y(-\xi)|$$

```
(%i2) tm_plot2d([1+0.2*x, 1/(1-0.2*x),
      exp(0.2*x)], [x, -%pi, %pi], [color,
      red, blue, green])$
```




For $t(x) = 1$ the oval degenerates to the original ellipse. Let's make a plot of the original ellipse alongside the oval resulting for each of the previous functions. The parameters will be set[b] for all plots to $(a, b, q) = (3, 2, 0.2)$.

```
(%i3) load(implicit_plot)$
(%i4) ip_grid: [100,100]$
(%i5) a: 3$ b: 2$ q: 0.2$
(%i8) O: x^2/a^2 + y^2*t/b^2 = 1$
(%i9) O0: O, t: 1$
(%i10) O1: O, t: 1+q*x$
(%i11) tm_implicit_plot([O0, O1], [x, -%pi,
       %pi], [y, -%pi, %pi], [color, black,
       red], [same_xy, true])$
```




```
(%i12) O2: O, t: 1/(1-q*x)$
(%i13) tm_implicit_plot([O0, O2], [x, -%pi,
       %pi], [y, -%pi, %pi], [color, black,
       blue], [same_xy, true])$
```




```
(%i14) O3: O, t: exp(q*x)$
```

[b] So we can reproduce the plots found at Mathematische Basteleien page.

[a] Web: konfou.xyz – Email: mail@konfou.xyz

(%i15) `tm_implicit_plot([O0, O3], [x, -%pi, %pi], [y, -%pi, %pi], [color, black, green], [same_xy, true])$`



And plotting them altogether

(%i16) `tm_implicit_plot([O0, O1, O2, O3], [x, -%pi, %pi], [y, -%pi, %pi], [color, black, red, blue, green], [same_xy, true])$`



As it can be seen the ovals do not differ by much. The reason behind this can become apparent by expanding to power series the perturbing functions from where it can be seen that those differ from the square term onward. Therefore the difference increases with $q^2$.

(%i17) `kill(q)$`

(%i18) `taylor(1+q*x, x, 0, 2)`

(%o18) $1+q\,x+\cdots$

(%i19) `taylor(1/(1-q*x), x, 0, 2)`

(%o19) $1+q\,x+q^2\,x^2+\cdots$

(%i20) `taylor(exp(q*x), x, 0, 2)`

(%o20) $1+q\,x+\dfrac{q^2\,x^2}{2}+\cdots$

Rather via an implicit equation, an oval can be described parametrically. The equations can be found by similarly perturbing the parametric equations of an ellipse so the symmetry to an axis is broken. An ellipse is parametrically defined

$$\mathbf{r}(\theta)=(a\cos\theta, b\sin\theta) \quad \theta\in[0,2\pi)$$

Then an oval can be described

$$\mathbf{r}(\theta)=(a\cos\theta, b\,\tau(\theta)\sin\theta\,) \quad a>b$$

By substituting to the equation at the beginning, that equation is converted in polar coordinates and takes the following form, from which we derive a relationship $\tau(\theta)=f(t(x))$.

$$\cos^2\theta+\sin^2\theta\tau^2(\theta)\,t(a\cos\theta)=1\Rightarrow\tau(\theta)=t^{-1/2}(a\cos\theta)$$

Therefore the oval's parametric equations can be written, for a similar result, utilizing the perturbing function $t(x)$ used in the beginning

$$\mathbf{r}(\theta)=(a\cos(\theta), b\sin(\theta)\,t^{-1/2}(a\cos\theta)) \quad a>b$$

Similar to before, let's make a plot of the ellipse alongside the oval resulting for each of the previous functions. The parameters will be set for all plots to $(a,b,q)=(3,2,0.2)$. The shapes are the same[c)] to the ones found previously.

(%i29) `a: 3$ b: 2$ q: 0.2$`

(%i32) `X: a*cos(u)$ Y: b*sin(u)/t^0.5$`

(%i34) `Y0: Y, t: 1$`

(%i35) `Y1: Y, t: 1+q*X$`

(%i36) `tm_plot2d([[parametric, X, Y0, [u, 0, 2*%pi]], [parametric, X, Y1, [u, 0, 2*%pi]]], [color, black, red], [same_xy, true])$`



(%i37) `Y2: Y, t: 1/(1-q*X)$`

c) An illusionary difference exists because the aspect ratios are not the same.

```
(%i38) tm_plot2d([[parametric, X, Y0, [u,
       0, 2*%pi]], [parametric, X, Y2, [u,
       0, 2*%pi]]], [color, black, blue],
       [same_xy, true])$
```




```
(%i39) Y3: Y, t: exp(q*X)$
(%i40) tm_plot2d([[parametric, X, Y0, [u,
       0, 2*%pi]], [parametric, X, Y3, [u,
       0, 2*%pi]]], [color, black, green],
       [same_xy, true])$
```




And plotting them altogether

```
(%i41) tm_plot2d([[parametric, X, Y0, [u, 0,
       2*%pi]], [parametric, X, Y1, [u, 0,
       2*%pi]], [parametric, X, Y2, [u, 0,
       2*%pi]], [parametric, X, Y3, [u, 0,
       2*%pi]]], [color, black, red, blue,
       green], [same_xy, true])$
```




Rather making the same shapes, a simpler, and more straightforward, alternative will be taking $\tau(\theta) = bt(c\cos\theta)$. The oval's parametric equations can then be written

$$\mathbf{r}(\theta) = (a\cos(\theta), \tau(\theta)\sin(\theta))$$

and the functions $\tau(\theta)$ will be $b + q\cos\theta$, $b(1 - q\cos\theta)^{-1}$, and $be^{q\cos\theta}$, where, for simplicity and uniformity, we've set the $\cos\theta$ scalar to just $q$. Plotting (setting for example $b=1$ and $q=0.2$) them it can be seen that

$$\forall x>0: |\tau(\theta(x))| \leqslant |\tau(\theta(-x))| \Rightarrow |y(\theta(x))| \geqslant |y(\theta(-x))|$$

```
(%i44) tm_plot2d([1+0.2*cos(u), 1/(1-
       0.2*cos(u)), exp(0.2*cos(u))], [u, 0,
       2*%pi], [color, red, blue, green])$
```




For $\tau(\theta) = b$ the oval degenerates to the original ellipse. Similar to before, let's make a plot of the ellipse alongside the oval resulting for each of the previous functions. The parameters will be set[d] for all plots to $(a, b, q) = (1, 0.7, 0.2)$.

```
(%i45) a: 1$ b: 0.7$ q: 0.2$
(%i48) X: a*cos(u)$ Y: t*sin(u)$
(%i50) Y0: Y, t: b$
(%i51) Y1: Y, t: b+q*cos(u)$
(%i52) tm_plot2d([[parametric, X, Y0, [u,
       0, 2*%pi]], [parametric, X, Y1, [u,
       0, 2*%pi]]], [color, black, red],
       [same_xy, true])$
```




```
(%i53) Y2: Y, t: b/(1-q*cos(u))$
```

[d] So we can reproduce the plot for the case $t = b + q\cos\theta$ given at Yamamoto's article no. V. At same article Yamamoto finds the parameters $(a, b, q) = (1, 0.72, 0.08)$ produce a curve that fits a real hen's egg exactly.

```
(%i54) tm_plot2d([[parametric, X, Y0, [u,
       0, 2*%pi]], [parametric, X, Y2, [u,
       0, 2*%pi]]], [color, black, blue],
       [same_xy, true])$
```




```
(%i55) Y3: Y, t: b*exp(q*cos(u))$
(%i56) tm_plot2d([[parametric, X, Y0, [u,
       0, 2*%pi]], [parametric, X, Y3, [u,
       0, 2*%pi]]], [color, black, green],
       [same_xy, true])$
```




And plotting them altogether

```
(%i58) tm_plot2d([[parametric, X, Y0, [u, 0,
       2*%pi]], [parametric, X, Y1, [u, 0,
       2*%pi]], [parametric, X, Y2, [u, 0,
       2*%pi]], [parametric, X, Y3, [u, 0,
       2*%pi]]], [color, black, red, blue,
       green], [same_xy, true])$
```




This strategy can easily be applied to the three dimensions. As opposed to starting from an ellipsoid, we can just start from a sphere. A sphere[6] with center at the origin may be specified in spherical coordinates by

$$\mathbf{r}(\theta,\phi) = (\cos\theta\sin\phi, \sin\theta\sin\phi, \cos\phi)$$

where $\theta\in[0,2\pi)$, $\phi\in[0,\pi]$. Then an oval shape, that is an egg, can be described[e] by

$$\mathbf{r}(\theta,\phi) = (t(\phi)\cos\theta\sin\phi, t(\phi)\sin\theta\sin\phi, a\cos\phi)$$

As in the beginning the variable $\phi$ isn't periodic so the perturbing function $t(\phi)$ shouldn't contain a periodic function either. So its form, similar to the beginning, can be $b+q\phi$, $b(1-q\phi)^{-1}$, and $be^{q\phi}$. The parameters will be set for all plots to $(a,b,q) = (1.8, 1, 0.2)$.

```
(%i59) a: 1.8$ b: 1$ q: 0.2$
(%i62) X: t*cos(u)*sin(v)$ Y:
       t*sin(u)*sin(v)$ Z: a*cos(v)$
(%i65) t1: b+q*v$ X1: X,t:t1$ Y1: Y,t:t1$
(%i68) tm_plot3d([X1, Y1, Z], [u, 0, 2*%pi],
       [v, 0, %pi], [same_xyz, true])$
```




```
(%i69) t2: 1/(1-q*v)$ X2: X,t:t2$ Y2: Y,
       t:t2$
(%i72) tm_plot3d([X2, Y2, Z], [u, 0, 2*%pi],
       [v, 0, %pi], [same_xyz, true])$
```




```
(%i73) t3: exp(q*v)$ X3: X,t:t3$ Y3: Y,t:t3$
```

[e] The same parametric equations can of course be attained rotating the previously found oval curves in space.

```
(%i76) tm_plot3d([X3, Y3, Z], [u, 0, 2*%pi],
       [v, 0, %pi], [same_xyz, true])$
```




Let's go back to two-dimensions. The inclined, that is neither parallel nor perpendicular, to the rotation axis plane section[f] of a cylinder is an ellipse. Then an oval can be revealed by the plane section of a cylinder deformed towards the one end. That surface is called pseudosphere[7][8] and, considering a cylinder has zero Gaussian curvature, it has negative curvature.

A pseudosphere is parametrically defined

$$\mathbf{r}(\theta,\phi)=\left(\cos\theta\sin\phi,\sin\theta\sin\phi,\cos\phi+\ln\left[\tan\frac{\phi}{2}\right]\right)$$

where $\theta\in[0,2\pi)$, $\phi\in[0,\pi)$. Suppose an inclined plane given implicitly

$$a x+c z+d=0\Rightarrow\alpha x+z+d=0$$

Let's start by plotting the pseudosphere and an intersecting plane.

```
(%i77) X: cos(u)*sin(v)$ Y: sin(u)*sin(v)$
       Z: cos(v)+log(tan(v/2))$
(%i80) tm_plot3d([X, Y, Z], [u, 0, 2*%pi],
       [v, 0, %pi], [same_xyz, true])$
```




By substituting the pseudosphere coordinates in the plane equation, we derive the implicit equation described the oval formed by the plane section.

$$\alpha\cos\theta\sin\phi+\cos\phi+\ln\left[\tan\frac{\phi}{2}\right]+d=0$$

[f]Plane section is called the intersection of a plane with a surface.

```
(%i81) a: 6.28$ d: 3.14$
(%i83) O: a*cos(u)*sin(v)-cos(u)-log(tan(v/
       2))+d=0
```

(%o83) $6.28\cos(u)\sin(v)-\log\left(\tan\left(\frac{v}{2}\right)\right)-\cos(u)+3.14=0$

```
(%i84) tm_implicit_plot(O, [u, 0, 2*%pi],
       [v, 0, %pi])$
```




Another surface of which planar sections give ovals are ring tori. This is an oval family that was defined geometrically and studied by Cassini. Unexpectedly, it was found that Cassini ovals[9][10] are cross-sections of a ring torus with planes parallel to the rotation axis of the torus.

Suppose the torus (having $x$ as rotation axis) with equation

$$(x^2+y^2+z^2+R^2-r^2)^2=4R^2(x^2+z^2)$$

and the plane $z=r$. Their intersection is

$$(x^2+y^2+R^2)^2=4R^2(x^2+r^2)$$

and after some algebra manipulation

$$(x^2+y^2)^2-2R^2(x^2-y^2)=4R^2r^2-R^4$$

which, with suitable of choice of parameters $b,a$, can be rewritten

$$(x^2+y^2)^2-2a^2(x^2-y^2)=b^4-a^4=a^4(e^4-1)$$

where $e=b/a$. Finally in polar coordinates, Cassini ovals are defined by the equation

$$r^4-2a^2r^2\cos 2\theta=a^4(e^4-1)$$

For $b>a\Leftrightarrow e>1$ the curve is an elliptical oval, for $b=a\Leftrightarrow e=1$ it becomes a lemniscate[g] (also known as lemniscate of Bernoulli after Bernoulli who studied this case), and for $b<a\Leftrightarrow e<1$ the curve divides to two ovals.

First let's plot[h] the lemniscate case.

```
(%i85) a: 1$ e: 1$
```

[g]Lemniscate is an umbrella term for figure-eight or $\infty$-shaped curves.

[h]Utilizing the polar equation and converting to $xy$ coordinates.

```
(%i87) 0: r^4-2*a^2*r^2*cos(2*u)=a^4*(e^4-
       1), r: sqrt(x^2 + y^2), u: atan(y/x)
```

(%o87) $(y^2+x^2)^2-2\,(y^2+x^2)\cos\left(2\arctan\left(\frac{y}{x}\right)\right)=0$

```
(%i88) tm_implicit_plot(0, [x, -2, 2], [y, -
       1, 1])$
```




Though interesting its pointy end makes this curve (closer to pyriform) unsuitable for describing the shape of an egg. Instead the end should be pointer than the other side but still smooth (that is an oval). Let's decrease to $e<1$ and plot the two disconnected ovals.

```
(%i89) a: 1$ e: 0.99205$
(%i91) 0: r^4-2*a^2*r^2*cos(2*u)=a^4*(e^4-
       1), r: sqrt(x^2 + y^2), u: atan(y/x)
```

(%o91) $(y^2+x^2)^2-2\,(y^2+x^2)\cos\left(2\arctan\left(\frac{y}{x}\right)\right)=-0.03142279084494404$

```
(%i92) tm_implicit_plot(0, [x, -2, 2], [y, -
       1.5, 1.5])$
```




Those shapes are similar to those seen in the rest of the article.

Kinda unrelated: few years ago, Aghekyan & Sahakyan proved a theorem that states the motion around the locus of the critical points of fourth-degree polynomials, having zeros that are pairwise symmetric with respect to the origin, are Cassini ovals.

## References


1. Weisstein, E. "Oval." From MathWorld. https://mathworld.wolfram.com/Oval.html
2. Crespi, S. and You, S. (2017). "Cracking the mystery of egg shape." From Science. https://vis.sciencemag.org/eggs/
3. Köller, J. and contributors. "Egg Curves and Ovals." http://www.mathematische-basteleien.de/eggcurves.htm
4. Yamamoto, N. "Equation of Egg Shaped Curve." http://nyjp07.com/index_egg_E.html
5. Weisstein, E. "Ellipse." From MathWorld. https://mathworld.wolfram.com/Ellipse.html
6. Weisstein, E. "Sphere." From MathWorld. https://mathworld.wolfram.com/Sphere.html
7. Weisstein, E. "Pseudosphere." From MathWorld. https://mathworld.wolfram.com/Pseudosphere.html
8. Wikipedia contributors. "Pseudosphere." https://en.wikipedia.org/wiki/Pseudosphere
9. Weisstein, E. "Cassini Ovals." From MathWorld. https://mathworld.wolfram.com/CassiniOvals.html
10. Wikipedia contributors. "Cassini oval." https://en.wikipedia.org/wiki/Cassini_oval
11. Aghekyan, G. and Sahakyan, K. (2015). "Cassini Ovals in Dynamic Geometry of Polynomials." http://www.ysu.am/files/UJAM1-12602974.pdf